introvoyz045's picture
Migrated from GitHub
4a56cd6 verified
Raw History Blame Contribute Delete
9.11 kB
import Mathlib
namespace OAI
namespace Alternating
open scoped BigOperators
open MeasureTheory
abbrev Instruction := β„• Γ— β„• Γ— Fin 3
abbrev Machine := β„• Γ— β„• Γ— Finset β„• Γ— List (List (Option Instruction))
abbrev MachineInput := Machine Γ— List β„•
def Machine.stateCount (M : Machine) : β„• := M.1 + 1
def Machine.symbolCount (M : Machine) : β„• := M.2.1 + 1
def Machine.haltingStates (M : Machine) : Finset β„• := M.2.2.1
def Machine.table (M : Machine) : List (List (Option Instruction)) := M.2.2.2
def Machine.WellFormed (M : Machine) : Prop :=
M.table.length = M.stateCount ∧
(βˆ€ q ∈ M.haltingStates, q < M.stateCount) ∧
βˆ€ row ∈ M.table,
row.length = M.symbolCount ∧
βˆ€ i : Instruction, some i ∈ row β†’
i.1 < M.stateCount ∧ i.2.1 < M.symbolCount
def ValidInput (I : MachineInput) : Prop :=
I.1.WellFormed ∧ βˆ€ a ∈ I.2, a < I.1.symbolCount
theorem ValidInput.machine_wellFormed {I : MachineInput} (hI : ValidInput I) :
I.1.WellFormed := hI.1
structure Configuration where
state : β„•
head : β„€
tape : β„€ β†’ β„•
def Machine.isHalting (M : Machine) (q : β„•) : Bool :=
decide (q = M.stateCount ∨ q ∈ M.haltingStates)
def Machine.instruction (M : Machine) (q a : β„•) : Option Instruction :=
((M.table[q]?).bind fun row => row[a]?).join
def Machine.step (M : Machine) (c : Configuration) : Configuration :=
if M.isHalting c.state then c
else match M.instruction c.state (c.tape c.head) with
| none => { c with state := M.stateCount }
| some i =>
{ state := i.1
head := c.head + (i.2.2.val : β„€) - 1
tape := Function.update c.tape c.head i.2.1 }
def initialConfiguration (w : List β„•) : Configuration :=
{ state := 0
head := 0
tape := fun j => if j < 0 then 0 else (w[j.toNat]?).getD 0 }
def configurationAt (I : MachineInput) (n : β„•) : Configuration :=
I.1.step^[n] (initialConfiguration I.2)
def Halts (I : MachineInput) : Prop :=
βˆƒ n : β„•, I.1.isHalting (configurationAt I n).state = true
abbrev Space := EuclideanSpace ℝ (Fin 3)
abbrev Field := ℝ β†’ Space β†’ Space
abbrev Pressure := ℝ β†’ Space β†’ ℝ
abbrev MultiIndex := Fin 3 β†’ β„•
noncomputable section
def basisVector (i : Fin 3) : Space := EuclideanSpace.single i 1
def multiOrder (Ξ± : MultiIndex) : β„• := βˆ‘ i, Ξ± i
def spatialD {E : Type*} [NormedAddCommGroup E] [NormedSpace ℝ E]
(i : Fin 3) (g : Space β†’ E) : Space β†’ E :=
fun x => fderiv ℝ g x (basisVector i)
def spatialPartial {E : Type*} [NormedAddCommGroup E] [NormedSpace ℝ E]
(Ξ± : MultiIndex) (g : Space β†’ E) : Space β†’ E :=
(spatialD 2)^[Ξ± 2] ((spatialD 1)^[Ξ± 1] ((spatialD 0)^[Ξ± 0] g))
def timeD {E : Type*} [NormedAddCommGroup E] [NormedSpace ℝ E]
(g : ℝ β†’ E) : ℝ β†’ E := fun t => derivWithin g (Set.Ici 0) t
def mixedPartial (u : Field) (l : β„•) (Ξ± : MultiIndex) (t : ℝ) (x : Space) : Space :=
timeD^[l] (fun s => spatialPartial Ξ± (u s) x) t
def timeDerivative (u : Field) (t : ℝ) (x : Space) : Space :=
timeD (fun s => u s x) t
def divergence (u : Space β†’ Space) (x : Space) : ℝ :=
βˆ‘ i : Fin 3, spatialD i u x i
def gradient (p : Space β†’ ℝ) (x : Space) : Space :=
(EuclideanSpace.equiv (Fin 3) ℝ).symm (fun i => spatialD i p x)
def laplacian (u : Space β†’ Space) (x : Space) : Space :=
βˆ‘ i : Fin 3, spatialD i (spatialD i u) x
def convection (u : Space β†’ Space) (x : Space) : Space :=
βˆ‘ i : Fin 3, (u x i) β€’ spatialD i u x
def residual (Ξ½ : ℝ) (u : Field) : Field :=
fun t x => timeDerivative u t x + convection (u t) x - Ξ½ β€’ laplacian (u t) x
def timeSpace : Set (ℝ Γ— Space) := Set.Ici 0 Γ—Λ’ Set.univ
def SmoothUpToZero (u : Field) : Prop :=
ContDiffOn ℝ (⊀ : β„•βˆž) (Function.uncurry u) timeSpace
def SupportedIn (K : Set Space) (u : Field) : Prop :=
βˆ€ t : ℝ, 0 ≀ t β†’ βˆ€ x : Space, x βˆ‰ K β†’ u t x = 0
def RapidDecay (u : Field) : Prop :=
βˆ€ (J l : β„•) (Ξ± : MultiIndex), βˆƒ C : ℝ, 0 ≀ C ∧
βˆ€ (t : ℝ), 0 ≀ t β†’ βˆ€ x : Space,
(1 + t + β€–xβ€–) ^ J * β€–mixedPartial u l Ξ± t xβ€– ≀ C
def ClassicalRegularity (u : Field) (p : Pressure) : Prop :=
(βˆ€ t : ℝ, 0 ≀ t β†’ ContDiff ℝ 2 (u t) ∧ ContDiff ℝ 1 (p t)) ∧
(βˆ€ x : Space, DifferentiableOn ℝ (fun t => u t x) (Set.Ici 0)) ∧
βˆ€ T : ℝ, 0 ≀ T β†’
ContinuousOn (Function.uncurry (timeDerivative u)) (Set.Icc 0 T Γ—Λ’ Set.univ) ∧
(βˆ€ Ξ± : MultiIndex, multiOrder Ξ± ≀ 2 β†’
ContinuousOn (fun tx : ℝ Γ— Space => spatialPartial Ξ± (u tx.1) tx.2)
(Set.Icc 0 T Γ—Λ’ Set.univ)) ∧
(βˆ€ Ξ± : MultiIndex, multiOrder Ξ± ≀ 1 β†’
ContinuousOn (fun tx : ℝ Γ— Space => spatialPartial Ξ± (p tx.1) tx.2)
(Set.Icc 0 T Γ—Λ’ Set.univ))
def NSSolution (Ξ½ : ℝ) (f u : Field) (p : Pressure) : Prop :=
ClassicalRegularity u p ∧
(βˆ€ x : Space, u 0 x = 0) ∧
βˆ€ t : ℝ, 0 ≀ t β†’ βˆ€ x : Space,
divergence (u t) x = 0 ∧
timeDerivative u t x + convection (u t) x =
-gradient (p t) x + Ξ½ β€’ laplacian (u t) x + f t x
abbrev SpatialL2 (E : Type*) [NormedAddCommGroup E] :=
MeasureTheory.Lp E 2 (MeasureTheory.volume : MeasureTheory.Measure Space)
def ContinuousInSobolev {E : Type*} [NormedAddCommGroup E] [NormedSpace ℝ E]
(k : β„•) (g : ℝ β†’ Space β†’ E) (T : ℝ) : Prop :=
βˆ€ Ξ± : MultiIndex, multiOrder Ξ± ≀ k β†’
βˆƒ G : ℝ β†’ SpatialL2 E,
ContinuousOn G (Set.Icc 0 T) ∧
βˆ€ t ∈ Set.Icc (0 : ℝ) T,
(fun x : Space => (G t) x) =ᡐ[volume] spatialPartial α (g t)
def C1InL2 (u : Field) (T : ℝ) : Prop :=
βˆƒ U W : ℝ β†’ SpatialL2 Space,
ContinuousOn U (Set.Icc 0 T) ∧ ContinuousOn W (Set.Icc 0 T) ∧
βˆ€ t ∈ Set.Icc (0 : ℝ) T,
((fun x : Space => (U t) x) =ᡐ[volume] u t) ∧
((fun x : Space => (W t) x) =ᡐ[volume] timeDerivative u t) ∧
HasDerivWithinAt U (W t) (Set.Icc 0 T) t
def EnergyClass (u : Field) (p : Pressure) : Prop :=
βˆ€ T : ℝ, 0 ≀ T β†’
ContinuousInSobolev 2 u T ∧ C1InL2 u T ∧ ContinuousInSobolev 1 p T ∧
βˆƒ C : ℝ, βˆ€ t ∈ Set.Icc (0 : ℝ) T, βˆ€ x : Space,
β€–u t xβ€– + β€–fderiv ℝ (u t) xβ€– ≀ C
def CompetitorEnergyClass (u : Field) (p : Pressure) : Prop :=
βˆ€ T : ℝ, 0 ≀ T β†’
ContinuousInSobolev 2 u T ∧ C1InL2 u T ∧ ContinuousInSobolev 1 p T ∧
βˆƒ C : ℝ, βˆ€ t ∈ Set.Icc (0 : ℝ) T, βˆ€ x : Space, β€–u t xβ€– ≀ C
def SolvesParticle (u : Field) (a : Space) (Ξ³ : ℝ β†’ Space) : Prop :=
Ξ³ 0 = a ∧ βˆ€ t : ℝ, 0 ≀ t β†’
HasDerivWithinAt Ξ³ (u t (Ξ³ t)) (Set.Ici 0) t
def GlobalMaterialFlow (u : Field) (X : ℝ β†’ Space β†’ Space) : Prop :=
(βˆ€ a : Space, SolvesParticle u a (fun t => X t a)) ∧
βˆ€ (a : Space) (Ξ³ : ℝ β†’ Space), SolvesParticle u a Ξ³ β†’
βˆ€ t : ℝ, 0 ≀ t β†’ Ξ³ t = X t a
def observedParticle : Space := -basisVector 0
abbrev Program := Nat.Partrec.Code
abbrev RationalPoint := β„š Γ— β„š Γ— β„š
def rationalPoint (q : RationalPoint) : Space :=
(EuclideanSpace.equiv (Fin 3) ℝ).symm ![(q.1 : ℝ), (q.2.1 : ℝ), (q.2.2 : ℝ)]
def error (k : β„•) : ℝ := (2 : ℝ)⁻¹ ^ k
def ComputableReal (x : ℝ) : Prop :=
βˆƒ c : Program, βˆ€ k : β„•, βˆƒ q : β„š,
Encodable.encode q ∈ c.eval k ∧ |x - (q : ℝ)| ≀ error k
abbrev FieldProgram := Program Γ— Program Γ— Program
def EvaluatesField (c : FieldProgram) (u : Field) : Prop :=
(βˆ€ (l : β„•) (Ξ± : MultiIndex) (t : β„š), 0 ≀ t β†’
βˆ€ (x : RationalPoint) (k : β„•), βˆƒ q : RationalPoint,
Encodable.encode q ∈ c.1.eval (Encodable.encode (l, α, t, x, k)) ∧
β€–mixedPartial u l Ξ± (t : ℝ) (rationalPoint x) - rationalPoint qβ€– ≀ error k) ∧
(βˆ€ (l : β„•) (Ξ± : MultiIndex) (k : β„•), βˆƒ m : β„•,
m ∈ c.2.1.eval (Encodable.encode (l, α, k)) ∧
βˆ€ (t s : ℝ), 0 ≀ t β†’ 0 ≀ s β†’ βˆ€ x y : Space,
|t - s| + β€–x - yβ€– ≀ error m β†’
β€–mixedPartial u l Ξ± t x - mixedPartial u l Ξ± s yβ€– ≀ error k) ∧
(βˆ€ (J l : β„•) (Ξ± : MultiIndex), βˆƒ C : β„š,
Encodable.encode C ∈ c.2.2.eval (Encodable.encode (J, l, Ξ±)) ∧ 0 ≀ C ∧
βˆ€ t : ℝ, 0 ≀ t β†’ βˆ€ x : Space,
(1 + t + β€–xβ€–) ^ J * β€–mixedPartial u l Ξ± t xβ€– ≀ (C : ℝ))
def AlternatingOutput (Ξ½ : ℝ) (K : Set Space) (I : MachineInput)
(velocityProgram forceProgram : FieldProgram) : Prop :=
βˆƒ (u f : Field) (X : ℝ β†’ Space β†’ Space),
EvaluatesField velocityProgram u ∧ EvaluatesField forceProgram f ∧
SmoothUpToZero u ∧ SmoothUpToZero f ∧
SupportedIn K u ∧ SupportedIn K f ∧ RapidDecay u ∧ RapidDecay f ∧
f = residual ν u ∧
NSSolution ν f u (fun _ _ => 0) ∧ EnergyClass u (fun _ _ => 0) ∧
(βˆ€ (v : Field) (p : Pressure), NSSolution Ξ½ f v p β†’ CompetitorEnergyClass v p β†’
βˆ€ t : ℝ, 0 ≀ t β†’ βˆ€ x : Space, v t x = u t x ∧ p t x = 0) ∧
GlobalMaterialFlow u X ∧
((βˆƒ t : ℝ, 0 ≀ t ∧ 0 < X t observedParticle 0) ↔ Halts I)
end
end Alternating
end OAI